Maple-assisted proof of formula for A181256

Robert Israel

10 July 2018

There are 213 configurations for a 2×4 sub-array where each 2×3 block has at most four 1's. We encode these configurations as lists in the order

$$\begin{bmatrix} 1 & 2 & 3 & 4 \\ 5 & 6 & 7 & 8 \end{bmatrix}.$$

Consider the 213×213 transition matrix T with entries $T_{ij} = 1$ if the bottom two rows of a 3×4 sub-array could be in configuration i while the top two rows are in configuration j, and 0 otherwise. The following Maple code computes it.

```
> okconfig:= proc(L) andmap(i -> L[i+1]+L[i+2]+L[i+3]+L[i+5]+L[i+6]
  +L[i+7]<=4,[$0..1]) end proc:
  Configs:=  select(okconfig, [seq(convert(2^8+i,base,2)[1..8],i=0.
  .2^8-1)]):
> Compatible:= proc(i,j) local k;
   if Configs[i][1..4] <> Configs[j][5..8] then return 0 fi;
   if add(Configs[i][k],k=[1,2,3,5,6,7])+add(Configs[j][k],k=1..3)=
  4 and
      add(Configs[i][k],k=[2,3,4,6,7,8])+add(Configs[j][k],k=2..4)=
  4 then 1 else 0 fi
  end proc:
  T:= Matrix(213,213,Compatible):
```

Thus for $n \geq 1$ $\quad a(n) = v^T\, T^n v$ where v is a column vector with all entries 1.

```
> v:= Vector(213,1):
```

To check, here are the first few entries of our sequence.

```
> TV[0]:= v:
  for n from 1 to 23 do TV[n]:= T . TV[n-1] od:
> A:= [seq(v^%T . TV[n],n=1..23)];
```

$$A := [336, 746, 1684, 3942, 10348, 27554, 74784, 212570, 608476, 1755630, 5152996, 15171626, 44814936, 133231586, 396621844, 1182345462, 3533450908, 10565961074, 31613799504, 94685091050, 283660310476, 850017916350, 2548227410836] \tag{1}$$

Now here is the minimal polynomial P of T, as computed by Maple.

```
> P:= unapply(LinearAlgebra:-MinimalPolynomial(T, t), t);
```

$$P := t \mapsto t^{23} - 60\, t^{20} + 1298\, t^{17} - 13344\, t^{14} + 70209\, t^{11} - 187380\, t^8 + 234252\, t^5 - 104976\, t^2 \tag{2}$$

```
> degree(P(t));
```

$$23 \tag{3}$$

This turns out to have degree 23, but with the t^0 and t^1 coefficients 0. Thus we will have

$0 = u\, P(T)\, T^n\, v = \sum_{i=2}^{23} p_i\, b(i+n)$ where p_i is the coefficient of t^i in $P(t)$. That corresponds to a homogeneous linear recurrence of order 21, which would hold true for any u and v, after a delay of 2.

It seems that with our particular $u = v^T$ and v we have a recurrence of order only 17, corresponding to a factor of P .

```
> empirical:= a(n)=4*a(n-1)-3*a(n-2)+32*a(n-3)-128*a(n-4)+96*a(n-5)
  -375*a(n-6)+1500*a(n-7)-1125*a(n-8)+1980*a(n-9)-7920*a(n-10)
  +5940*a(n-11)-4644*a(n-12)+18576*a(n-13)-13932*a(n-14)+3888*a
  (n-15)-15552*a(n-16)+11664*a(n-17):
  Q:= unapply(add(coeff((lhs-rhs)(empirical),a(n-i))*t^(17-i),i=0.
  .17),t);
```

$$Q := t \mapsto t^{17} - 4\,t^{16} + 3\,t^{15} - 32\,t^{14} + 128\,t^{13} - 96\,t^{12} + 375\,t^{11} - 1500\,t^{10} + 1125\,t^{9} - 1980\,t^{8} + 7920\,t^{7} - 5940\,t^{6} + 4644\,t^{5} - 18576\,t^{4} + 13932\,t^{3} - 3888\,t^{2} + 15552\,t - 11664 \quad \textbf{(4)}$$

The complementary factor $R(t) = \dfrac{P(t)}{Q(t)}$ has degree 6, again with the lowest two coefficients 0.

```
> R:= unapply(normal(P(t)/Q(t)),t);
```

$$R := t \mapsto \left(t^4 + 4\,t^3 + 13\,t^2 + 12\,t + 9\right)\,t^2 \quad \textbf{(5)}$$

```
> degree(R(t));
```

$$6 \quad \textbf{(6)}$$

Now we want to show that $c(n) = v^T\,Q(T)\,T^n\,v = 0$ for all $n \geq 1$. This will certainly satisfy the recurrence

$$\sum_{i=2}^{6} r_i\,c(i+n) = \sum_{i=2}^{6} r_i u\,Q(T)\,T^{n+i} v = u\,Q(T)\,R(T)\,T^n\,v = u\,P(T)\,T^n\,v = 0$$

where r_i are the coefficients of $R(t)$. To show all $c(n) = 0$ it suffices to show $c(1) = ... = c(5) = 0$.

```
> UT[0]:= v^%T:
  for n from 1 to 17 do UT[n]:= UT[n-1].T od:
  w:= add(coeff(Q(t),t,j)*UT[j],j=0..17):
> seq(w . TV[n],n=1..5);
```

$$0, 0, 0, 0, 0 \quad \textbf{(7)}$$

This completes the proof.